Lezione 1

```
(* la riduzione dei polinomi in forma normale è automatica *)
```

In[1]:=

```
4/3 a + b c + a b a^2 + 2/3 a - b c
```

Out[1]=

$$2\,a + a^3\,b$$

```
(* si può evitare con HoldForm *)
```

In[2]:=

```
HoldForm[4/3 a + b c + a b a^2 + 2/3 a - b c]
```

Out[2]=

$$\frac{4\,a}{3} + b\,c + a\,b\,a^2 + \frac{2\,a}{3} - b\,c$$

```
(* i prodotti di polinomi non vengono eseguiti automaticamente *)
```

In[3]:=

```
(2 x + 4 a) (x - a)^3 (x^2 - 5/2 a x + 3/4)
```

Out[3]=

$$(-a + x)^3\,(4\,a + 2\,x)\left(\frac{3}{4} - \frac{5\,a\,x}{2} + x^2\right)$$

In[4]:=

```
Expand[%] (* ... si possono ottenere con Expand *)
```

Out[4]=

$$-3\,a^4 + \frac{15\,a^3\,x}{2} + 10\,a^5\,x - \frac{9\,a^2\,x^2}{2} - 29\,a^4\,x^2 - \frac{3\,a\,x^3}{2} + 25\,a^3\,x^3 + \frac{3\,x^4}{2} - a^2\,x^4 - 7\,a\,x^5 + 2\,x^6$$

In[5]:=

```
Factor[%] (* per fattorizzare a coefficienti interi *)
```

Out[5]=

$$\frac{(a - x)^3\,(2\,a + x)\,(-3 + 10\,a\,x - 4\,x^2)}{2}$$

In[6]:=

```
(1 + (Sin[x] + Cos[x])^2)/(1 + Sin[x] Cos[x])
```

Out[6]=

$$\frac{1 + (\mathrm{Cos}[x] + \mathrm{Sin}[x])^2}{1 + \mathrm{Cos}[x]\,\mathrm{Sin}[x]}$$

In[7]:=

```
Factor[%]
```

Out[7]=

$$\frac{1 + \text{Cos}[x]^2 + 2\,\text{Cos}[x]\,\text{Sin}[x] + \text{Sin}[x]^2}{1 + \text{Cos}[x]\,\text{Sin}[x]}$$

In[8]:=

```
Simplify[%] (* per le semplificazioni *)
```

Out[8]=

```
2
```

```
(* Non sempre le cose vanno come ci si aspetta ... *)
```

In[9]:=

```
(2(x - 1)^8 - 2)/(2 x)
```

Out[9]=

$$\frac{-2 + 2\,(-1 + x)^8}{2\,x}$$

In[10]:=

```
Simplify[%]
```

Out[10]=

$$\frac{-2 + 2\,(-1 + x)^8}{2\,x}$$

In[11]:=

```
(Sqrt[6] + Sqrt[2]/3)^2
```

Out[11]=

$$\left(\frac{\text{Sqrt}[2]}{3} + \text{Sqrt}[6]\right)^2$$

In[12]:=

```
Expand[%]
```

Out[12]=

$$\frac{56}{9} + \frac{4}{\text{Sqrt}[3]}$$

In[13]:=

```
N[%] (* per ottenere il valore numerico *)
```

Out[13]=

```
8.53162
```

In[14]:=

```
% - N[4/Sqrt[3]]
```

Out[14]=

```
6.22222
```

In[15]:=
```
Rationalize[%] (* per tornare alla forma razionale *)
```
Out[15]=
$$\frac{56}{9}$$

In[16]:=
```
V = {vx,vy,vz} (* i vettori sono rappresentati come liste *)
W = {wx,wy,wz}
```
Out[16]=
```
{vx, vy, vz}
```
Out[17]=
```
{wx, wy, wz}
```
In[18]:=
```
3 V + W
```
Out[18]=
```
{3 vx + wx, 3 vy + wy, 3 vz + wz}
```
In[19]:=
```
V . W (* per il prodotto scalare *)
```
Out[19]=
```
vx wx + vy wy + vz wz
```
In[20]:=
```
M = {{a,b},{c,d}} (* una matrice è una lista di liste (righe) *)
```
Out[20]=
```
{{a, b}, {c, d}}
```
In[21]:=
```
MatrixForm[%] (* per visualizzare le matrici in righe e colonne *)
```
Out[21]//MatrixForm=
$$\begin{matrix} a & b \\ c & d \end{matrix}$$

In[22]:=
```
M[[2,1]] (* l'elemento di indici 2,1 *)
```
Out[22]=
```
c
```
In[23]:=
```
M . {x,y} (* prodotto matrice vettore *)
```
Out[23]=
```
{a x + b y, c x + d y}
```

In[24]:=

```
MatrixForm[M . M] (* prodotto tra matrici *)
```

Out[24]//MatrixForm=

$$\begin{pmatrix} a^2 + b\,c & a\,b + b\,d \\ a\,c + c\,d & b\,c + d^2 \end{pmatrix}$$

In[25]:=

```
Det[%]
```

Out[25]=

$$b^2\,c^2 - 2\,a\,b\,c\,d + a^2\,d^2$$

In[26]:=

```
Factor[%]
```

Out[26]=

$$(-(b\,c) + a\,d)^2$$

In[27]:=

```
x = Random[Integer,10] (* assegnazione con valutazione immediata ... *)
```

Out[27]=

```
2
```

In[28]:=

```
X = {x,x,x,x,x}
```

Out[28]=

```
{2, 2, 2, 2, 2}
```

In[29]:=

```
y := Random[Integer,10] (* ... e con valutazione differita *)
```

In[30]:=

```
Y = {y,y,y,y,y}
```

Out[30]=

```
{0, 0, 10, 6, 7}
```

In[31]:=

```
Map[Sqrt,Y] (* Map per applicare una funzione ad
               ogni elemento di una lista *)
```

Out[31]=

```
{0, 0, Sqrt[10], Sqrt[6], Sqrt[7]}
```

In[32]:=

```
Apply[Max,Y] (* Apply per applicare una funzione
                ad una lista di argomenti *)
```

Out[32]=

```
10
```

In[33]:=

```
valoricasuali = Table[Random[Real,{0,5}],{i,1,1000}];
```

In[34]:=

```
ListPlot[valoricasuali]
```



Out[34]=

-Graphics-

In[35]:=

```
ListPlot[Sort[valoricasuali]]
```



Out[35]=

-Graphics-

In[36]:=

```
Clear[x,y] (* per riutilizzare x e y come simboli senza valore *)

(* La funzione Solve dà la lista delle soluzioni di un'equazione
   ogni soluzione è a sua volta espressa come una lista di regole
   di sostituzione *)
```

In[37]:=

```
Solve[a x^2 + b x + c == 0,x]
```

Out[37]=

$$\{\{x \to \frac{-b - \text{Sqrt}[b^2 - 4\,a\,c]}{2\,a}\}, \{x \to \frac{-b + \text{Sqrt}[b^2 - 4\,a\,c]}{2\,a}\}\}$$

In[38]:=

```
sol1 = %[[1]] (* la prima soluzione ... *)
```

Out[38]=

$$\{x \to \frac{-b - \text{Sqrt}[b^2 - 4\,a\,c]}{2\,a}\}$$

In[39]:=

```
sol2 = %%[[2]] (* ... la seconda che hai detto *)
```

Out[39]=

$$\{x \to \frac{-b + \text{Sqrt}[b^2 - 4\,a\,c]}{2\,a}\}$$

In[40]:=

```
a x^2 + b x + c == 0 /. sol1 (* sostituzione per la verifica *)
```

Out[40]=

$$c + \frac{b\,(-b - \text{Sqrt}[b^2 - 4\,a\,c])}{2\,a} + \frac{(-b - \text{Sqrt}[b^2 - 4\,a\,c])^2}{4\,a} == 0$$

In[41]:=

```
Simplify[%]
```

Out[41]=

```
True
```

In[42]:=

```
x1 = x /. sol1 (* dalla regola di sostituzione all'espressione *)
```

Out[42]=

$$\frac{-b - \text{Sqrt}[b^2 - 4\,a\,c]}{2\,a}$$

In[43]:=

```
Reduce[a x^2 + b x + c == 0,x] (* ecco la famigerata discussione ... *)
```

Out[43]=

$$a \neq 0 \;\&\&\; x == \frac{-b - \text{Sqrt}[b^2 - 4\,a\,c]}{2\,a} \;||$$

$$a \neq 0 \;\&\&\; x == \frac{-b + \text{Sqrt}[b^2 - 4\,a\,c]}{2\,a} \;||$$

$$c == 0 \;\&\&\; b == 0 \;\&\&\; a == 0 \;||\; b \neq 0 \;\&\&\; x == -\left(\frac{c}{b}\right) \;\&\&\; a == 0$$

In[44]:=

```
FindRoot[E^x == 1/x,{x,1}] (* ... un'equazione trascendente *)
```

Out[44]=

```
{x -> 0.567143}
```

In[45]:=

```
Plot[{E^x,1/x},{x,0.1,1}]
```



Out[45]=

-Graphics-

In[46]:=

```
dati = Table[{x,x - 10/x + Random[Real,{-.5,.5}]},{x,1,10,.05}];
```

In[47]:=

```
punti = ListPlot[dati]
```



Out[47]=

-Graphics-

In[48]:=

```
f1 = Fit[dati,{1},x] (* media *)
```

Out[48]=

2.91572

In[49]:=

```
f2 = Fit[dati,{1,x},x] (* retta di regressione *)
```

Out[49]=

-5.98757 + 1.61878 x

In[50]:=

```
f3 = Fit[dati,{1/x,1,x},x]
```

Out[50]=

$$-0.09363 - \frac{9.94105}{x} + 1.01256\ x$$

In[51]:=

```
f4 = Fit[dati,{1/x,x},x]
```

Out[51]=

$$\frac{-10.0767}{x} + 1.00233\ x$$

In[52]:=

```
Plot[{f1,f2,f4},{x,1,10}]
```



Out[52]=

-Graphics-

In[53]:=

```
Show[%,punti]
```



Out[53]=

-Graphics-

```
(* Come arrivare il più lontano possibile ... *)
```

In[54]:=

```
DSolve[{x''[t] == a,
        x[0] == x0,
        x'[0] == v0},x[t],t]
```

Out[54]=

$$\{\{x[t] \to \frac{a\ t^2}{2} + t\ v0 + x0\}\}$$

In[55]:=

```
x[t] /. %[[1]]
```

Out[55]=

$$\frac{a\ t^2}{2} + t\ v0 + x0$$

In[56]:=

```
% /. {a -> {0,-g},x0 -> {0,0}, v0 -> v {Cos[c],Sin[c]}}
```

Out[56]=

$$\{t\ v\ \text{Cos}[c],\ \frac{-(g\ t^2)}{2} + t\ v\ \text{Sin}[c]\}$$

In[57]:=

```
Solve[%[[2]] == 0,t]
```

Out[57]=

$$\{\{t \to 0\},\ \{t \to \frac{2\ v\ \text{Sin}[c]}{g}\}\}$$

In[58]:=

```
%[[2]]
```

Out[58]=

$$\{t \to \frac{2\ v\ \text{Sin}[c]}{g}\}$$

In[59]:=

```
%%% /. %
```

Out[59]=

$$\{\frac{2\ v^2\ \text{Cos}[c]\ \text{Sin}[c]}{g},\ 0\}$$

In[60]:=

```
Simplify[%[[1]]]
```

Out[60]=

$$\frac{v^2\ \text{Sin}[2\ c]}{g}$$

In[61]:=

```
Table[
  Block[{g = 9.81,v = 10,c = n Degree},
        ParametricPlot[%%%%%//Evaluate,{t,0,t /. %%%},
          AspectRatio->Automatic,PlotRange->{{0,12},{0,6}}]],
  {n,5,85,5}]
```




0 1 2 3 4 5 6
2 4 6 8 10 12


0 1 2 3 4 5 6
2 4 6 8 10 12


0 1 2 3 4 5 6
2 4 6 8 10 12


0 1 2 3 4 5 6
2 4 6 8 10 12


0 1 2 3 4 5 6
2 4 6 8 10 12


0 1 2 3 4 5 6
2 4 6 8 10 12


6
5
4
3
2
1
0
2
4
6
8
10
12


6
5
4
3
2
1
0
2
4
6
8
10
12


6
5
4
3
2
1
0
2
4
6
8
10
12


6
5
4
3
2
1
0
2
4
6
8
10
12


6
5
4
3
2
1
0
2
4
6
8
10
12


6
5
4
3
2
1
0
2
4
6
8
10
12



Out[61]=

```
{-Graphics-, -Graphics-, -Graphics-, -Graphics-, -Graphics-,
  -Graphics-, -Graphics-, -Graphics-, -Graphics-, -Graphics-,
  -Graphics-, -Graphics-, -Graphics-, -Graphics-, -Graphics-,
  -Graphics-, -Graphics-}
```

In[62]:=

```
Show[%]
```



Out[62]=

```
-Graphics-
```